The Why of Y or The Meaning of Recursion or The meaning of life

Eli Barzilay

eli@cs.cornell.edu

April 27, 1999

CS212: The Why of Y

Tiny language?

We know that it is possible to express: • numbers, • conditionals & booleans, • cons cells & lists. How far can we go?

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 1/38

Lambda Calculus

Consider Lambda-Calulus (the “spiritual” origin of Scheme). This is a formal language that have only the following expressions available: 1. Variables: x, y, z, . . .; 2. Lambda expressions: λx.E; 3. Application of expressions: E_1E_2.

For example, the definition of kons from section:

(define (kons x y) (lambda (s) (s x y)))

is written as: λx.λy.λs.sxy

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 2/38

Our version of the Lambda Calculus

Some things look disturbing:

• The syntax is very terse — very hard to read, • Only uses Lambda expressions with a single argument, • Nothing else besides these expressions! So we will fix these by the following:

• Use a Scheme-like syntax for our Lambda expressions;

• Use the fact that we can transform a program to a Curried-form using simple transformations, therefore use normal argument lists;

• Assume that we have predefined implementation of all basic things we need such as numbers etc.

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 3/38

Guessing game...

We still have a very important feature that we have in Scheme and don’t have in the Lambda Calculus.

? ? ?

Try to guess what.

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 4/38

No names!

What we’re missing is the ability to define variables!

(At least not global variables.)

It looks like it is impossible to program anything real if we don’t use symbols. Specifically, what can we do if there is no way to use free variables and bind such variables globally.

(This will get clearer shortly.)

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 5/38

No recursion!

Also related to the way that we introduced the Scheme

define and letrec special forms in the days of the

substitution model:

• we waved our hands a lot at these forms, • we gave you several versions of the evaluation rule, • it was always either too vague or too complicated. Even with the environment model, we used set! for translating letrec, and we used side-effects for defining variables.

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 6/38

No hope!

This is all trying to cover up the fact that recursion as we know it now, can only be used by a function that mentions its own name.

Is it impossible to use recursion without any form of side-effects?

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 7/38

First example:

Lets start now with with an example: a simple recursive function definition:

(define fact
  (lambda (n)
    (if (= n 0)
      1
      (* n (fact (- n 1))))))

When we look at the value of fact, we see that by itself, it doesn’t make any sense because fact is a free variable within the body:

(lambda (n)
  (if (= n 0)
    1
    (* n (fact (- n 1)))))

So, how is it possible to write the factorial function?

(or maybe it isn’t possible?)

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 8/38

Mathematicl example:

This is similar to mathematical definitions — you do not have any real way to name object, names are just short hand for what they stand for. For example, when you define:

A_0 ≡ 1 ∀n > 0 : A_n ≡ n · A_{n−1} you actually mean that you have this infinite list of definitions:

A_0 ≡ 1 A_1 ≡ 1 · A_0 = 1 · 1 = 1 A_2 ≡ 2 · A_1 = 2 · 1 = 2 A_3 ≡ 3 · A_2 = 3 · 2 = 6 .

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 9/38

Induction & Recursion

Induction is the tool that we use to show that the above definition is actually defined for all of the natural numbers.

But when you do such an inductive proof, you are actually using name of the function as if it is already defined — in the proof itself.

This is the same problem that we ran into with Lambda-Calculus.

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 10/38

Convenient syntax

To work on this we will continue using names — only use them as shortcuts for other forms, disallowing recursive definitions. We will use the “:=” metasyntax to emphasize this, so when we write this:

F := (lambda (x) x)
G := (lambda (y) (F y))
(G F)

it is just shortcut for what we actually mean and were too lazy to write:

((lambda (y) ((lambda (x) x) y)) (lambda (x) x))

but this:

F := (lambda (x) (F x))

is meaningless since it stands for an infinite expression, and that is impossible.

[Note: This syntax can be used in Scheme by just using defines for the “:=” assignments, if you are careful to to have any function that references itself.]

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 11/38

Start working on a solution

So lets begin to try writing the above factorial function:

fact :=
  (lambda (n)
    (if (= n 0)
      1
      (* n (fact (- n 1)))))

However, we cannot use a recursive call, so we can write anything we want instead of the “fact”, just to make it valid. For example, use 666:

fact :=
  (lambda (n)
    (if (= n 0)
      1
      (* n (666 (- n 1)))))

(Note that 666 will be a function in pure Lambda Calculus.)

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 12/38

When does it work?

This function will not work in the general case, but there is one case where it will work: when n=0 (since then we do not try to do that bogus call).

We can note this fact by calling this “fact0” instead:

fact0 :=
  (lambda (n)
    (if (= n 0)
      1
      (* n (666 (- n 1)))))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 13/38

Making it work for even more inputs

Now that we have that, we can use it to write “fact1” which is the same as our desired factorial for arguments of 0 and 1:

fact1 :=
  (lambda (n)
    (if (= n 0)
      1
      (* n (fact0 (- n 1)))))

Remeber that this is actually shorthand for:

fact1 :=
  (lambda (n)
    (if (= n 0)
      1
      (* n ((lambda (n)
              (if (= n 0)
                1
                (* n (666 (- n 1)))))
            (- n 1)))))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 14/38

And even more...

We can continue in this way and write “fact2” that will work for n<3:

fact2 :=
  (lambda (n)
    (if (= n 0)
      1
      (* n (fact1 (- n 1)))))

or, in full form:

fact2 :=
  (lambda (n)
    (if (= n 0)
      1
      (* n ((lambda (n)
              (if (= n 0)
                1
                (* n ((lambda (n)
                        (if (= n 0)
                          1
                          (* n (666 (- n 1)))))
                      (- n 1)))))
            (- n 1)))))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 15/38

Is there and end to this...

If we continue this way we will get the true fact function,

But: the problem is that to handle any possible integer argument, it will still have to be an infinite one:

Here is what it is supposed to look like:

fact0 := (lambda (n) (if (= n 0) 1 (* n (666 (- n 1)))))
fact1 := (lambda (n) (if (= n 0) 1 (* n (fact0 (- n 1)))))
fact2 := (lambda (n) (if (= n 0) 1 (* n (fact1 (- n 1)))))
fact3 := (lambda (n) (if (= n 0) 1 (* n (fact2 (- n 1)))))
...

And our fact is the ‘∞-th’ fact, having an infinite size.

— back to the original problem...

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 16/38

A little hope

We can see a faint beam of light:

This bigger and bigger definition uses multiple instances of the same original fact code, so we can try to abstract this away — pull the value that is being used as the internal call as an argument to a function.

Rule#1:

(... x0 ...)

⇕

((lambda (x) (... x ...)) x0)

— making sure that no variables get captured.

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 17/38

Use Rule#1

fact1 now becomes:

  fact1 :=
    ((lambda (fact)
       (lambda (n)
         (if (= n 0)
           1
           (* n (fact (- n 1))))))
     fact0)

Which is actually:

fact1 :=
  ((lambda (fact)
     (lambda (n)
       (if (= n 0)
         1
         (* n (fact (- n 1))))))
   ((lambda (fact)
      (lambda (n)
        (if (= n 0)
          1
          (* n (fact (- n 1))))))
    666))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 18/38

Make it look even better

This way we do have something that looks better, but we still repeat ourselves.

To solve this problem, we’ll use a function that will take the (lambda (n) ...) expression and will apply that on 666.

Rule#2:

(f x)

⇕

((lambda (g) (g x)) f)

— actually, an instance of Rule#1.

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 19/38

Use Rule#2

Use this, creating a function that gets make-fact:

fact0 :=
  ((lambda (make-fact)
     (make-fact 666))
   (lambda (fact)
     (lambda (n)
       (if (= n 0)
         1
         (* n (fact (- n 1)))))))

Now fact1 can be written easily:

fact1 :=
  ((lambda (make-fact)
     (make-fact (make-fact 666)))
   (lambda (fact)
     (lambda (n)
       (if (= n 0)
         1
         (* n (fact (- n 1)))))))

And the principle is clear.

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 20/38

Back to the “real” fact

We can now continue by working on the “real” fact, which is still infinite at this stage:

fact :=
  ((lambda (make-fact)
     (make-fact
      (make-fact
       (... (make-fact 666)...))))
   (lambda (fact)
     (lambda (n)
       (if (= n 0)
         1
         (* n (fact (- n 1)))))))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 21/38

Continue please...

Now, examine the (lambda (make-fact) ...) — all it does is get something and apply it to the result of applying it to the result of applying it it ... to 666.

An idea that we can use instead is to pass make-fact itself to make-fact, and then have it do the extra calls when needed.

Now the second closure gets make-fact itself, so the internal fact variable is renamed to make this clearer, and we initially apply it on 666:

fact :=
  ((lambda (make-fact)
     (make-fact make-fact))
   (lambda (make-fact)
     (lambda (n)
       (if (= n 0)
         1
         (* n ((make-fact 666)
               (- n 1)))))))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 22/38

Still problems ... and a solution!

That will make this function do the same as fact1 — if we try to continue, we’ll bump into 666, a reduction will demonstrate this (do one)...

Instead — use make-fact instead of 666 and we’ll be able to do as many calls as needed:

fact :=
  ((lambda (make-fact)
     (make-fact make-fact))
   (lambda (make-fact)
     (lambda (n)
       (if (= n 0)
         1
         (* n ((make-fact make-fact)
               (- n 1)))))))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 23/38

Voila! The ZF-100!

Now, we have something which is fact (convince by fixing the reductions made with 666).

So we see that it is possible to write recursive functions using finite expressions! — But we still have problems to overcome...

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 24/38

What problems?

First, there is still the problem of having a solution which is quite different from the original factorial function.

To make things more clear, we use more abstractions.

First, abstract the second lambda expression, putting the (make-fact make-fact) call outside the expression (Rule#1) so we get the fact’s original body in one piece:

fact :=
  ((lambda (make-fact)
     (make-fact make-fact))
   (lambda (make-fact)
     ((lambda (fact)
        (lambda (n)
          (if (= n 0)
            1
            (* n (fact (- n 1))))))
      (make-fact make-fact))))

Still, we have more problems...

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 25/38

More problems??

The problem is that if you will type this in — you’ll never see the prompt again...

This is because we use applicative-order evaluation, or eager evaluation — the rule that says that you first evaluate the function and its arguments, then do the apply step.

What we just did is wrote an expression that looks like:

((lambda (x) (x x))
 (lambda (x) (f (x x))))

And this is a variation on the well-known and much loved expression:

((lambda (x) (x x)) (lambda (x) (x x)))

...which evaluates to itself...

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 26/38

((λx.xx)(λx.xx))

this is the key for creating a loop — we use it to create the recursion. The original expression evaluates as follows:

((lambda (x) (x x)) (lambda (x) (f (x x))))
((lambda (x) (f (x x))) (lambda (x) (f (x x))))
(f ((lambda (x) (f (x x))) (lambda (x) (f (x x)))))
(f (f ((lambda (x) (f (x x))) (lambda (x) (f (x x))))))
(f (f (f ((lambda (x) (f (x x))) (lambda (x) (f (x x)))))))
.

The problem is that we must evaluate the argument to

f before evaluating the application of f to avoid an

infinite loop.

The solution for this is to use this rule:

Rule#3:

f

⇕

(lambda (z) (f z))

— as long as f is a one-arg function.

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 27/38

Use Rule#3

So we will use this wrapper on the “(x x)”:

((lambda (x) (x x))
 (lambda (x) (f (lambda (z) ((x x) z)))))

and when we evaluate this we get to something similar to this:

(f (lambda (z)
     (((lambda (x) (f (x x)))
       (lambda (x) (f (x x))))
      z)))

Except that we have (lambda (z) ((x x) z)) instead of the original (x x) so this is really:

(f (lambda (z)
     (((lambda (x)
         (f (lambda (z) ((x x) z))))
       (lambda (x)
         (f (lambda (z) ((x x) z)))))
      z)))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 28/38

Some more

... now the loop is deferred inside the lambda expression — and when we apply this value the result is still the same.

Note that this wouldn’t be a problem if we would use normal-order evaluation.

We can also note that nothing changes if we modify the first part of:

((lambda (x) (x x))
 (lambda (x) (f (lambda (z) ((x x) z)))))

to be identical with the second part — the evaluation still looks the same except for skipping the first step now:

((lambda (x) (f (lambda (z) ((x x) z))))
 (lambda (x) (f (lambda (z) ((x x) z)))))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 29/38

Back to the factorial

Applying the first fix to the above definition gives:

fact :=
  ((lambda (make-fact)
     (make-fact make-fact))
   (lambda (make-fact)
     ((lambda (fact)
        (lambda (n)
          (if (= n 0)
            1
            (* n (fact (- n 1))))))
      (lambda (z)
        ((make-fact make-fact) z)))))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 30/38

Yet more

And the second fix will make it:

fact :=
  ((lambda (make-fact)
     ((lambda (fact)
        (lambda (n)
          (if (= n 0)
            1
            (* n (fact (- n 1))))))
      (lambda (z)
        ((make-fact make-fact) z))))
   (lambda (make-fact)
     ((lambda (fact)
        (lambda (n)
          (if (= n 0)
            1
            (* n (fact (- n 1))))))
      (lambda (z)
        ((make-fact make-fact) z)))))

This does look more complex — but it is actually simpler since the two parts are the same.

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 31/38

Make it simple again

Now take the “(lambda (fact) ...)” thing, and call it fooct:

fooct :=
  (lambda (fact)
    (lambda (n)
      (if (= n 0)
        1
        (* n (fact (- n 1))))))

fact :=
  ((lambda (make-fact)
     (fooct
      (lambda (z)
        ((make-fact make-fact) z))))
   (lambda (make-fact)
     (fooct
      (lambda (z)
        ((make-fact make-fact) z)))))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 32/38

Even better

Using Rule#1 make fooct a variable, call the result function make-real-fact since fact is now the result of applying it:

fooct :=
  (lambda (fact)
    (lambda (n)
      (if (= n 0)
        1
        (* n (fact (- n 1))))))

make-real-fact :=
  (lambda (fooct)
    ((lambda (make-fact)
       (fooct
        (lambda (z)
          ((make-fact make-fact) z))))
     (lambda (make-fact)
       (fooct
        (lambda (z)
          ((make-fact make-fact) z))))))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 33/38

And now: fact!

And now the real fact can be defined with:

fact := (make-real-fact fooct)

This works — and the whole thing is not recursive! (Actually it is, but it is not defined recursively).

The last thing to note is that make-fact is totally independent of fact, so we can define it as a general function:

make-recursive :=
  (lambda (f)
    ((lambda (x)
       (f (lambda (z) ((x x) z))))
     (lambda (x)
       (f (lambda (z) ((x x) z))))))

fact := (make-recursive fooct)

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 34/38

We have general recursion!

This also work for other recursive functions:

fib0 :=
  (lambda (fib)
    (lambda (n)
      (if (<= n 1)
        n
        (+ (fib (- n 1))
           (fib (- n 2))))))
fib := (make-recursive fib0)

And another example:

length :=
  (make-recursive
    (lambda (length)
      (lambda (l)
        (if (null? l)
          0
          (+ (length (cdr l)) 1)))))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35/38

The “Y” Combinator

Our “make-recursive” function is usually called the fixpoint operator or the Y combinator.

It looks really simple when using the normal-order version (remember: our version is the applicative-order one):

Y := (lambda (f)
      ((lambda (x) (f (x x)))
       (lambda (x) (f (x x)))))

And this all comes from the loop generated by:

((lambda (x) (x x)) (lambda (x) (x x)))

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 36/38

The core of looping

((λx.xx)(λx.xx)) is also the idea behind many deep mathematical facts. (as you will discover in future courses). As an example, follow the next rule:

I will say the next sentence twice:
  ‘‘I will say the next sentence twice’’.

(Note the usage of colon for the first and quotes for the second — what is the equivalent of that in Lambda-Calculus?)

Final note: Here is a function that returns itself (not its code):

(Y (lambda (f) (lambda (x) f)))

which is the same as:

(define (f) f)

in Scheme.

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 37/38

Didn’t have enough?

Final exercises for whoever is interested (and still conscious):

1. Write an expression that returns its own Sexpression when evaluated.

A similar problem with C and strings:

char *s="char *s=%c%s%c;%cmain()printf(s,34,s,34,10,10);%c";
main()printf(s,34,s,34,10,10);

2. Our make-recursive function can only work on oneargument functions. Why? Write a version for two arguments.

3. How can mutual recursion be implemented using this idea?

CS212: The Why of Y . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 38/38